Primitivní funkce

Primitivní funkce se v programu *Mathematica* získají pomocí neurčitého integrálu. Ten se obvykle dá programem spočítat, ale jsou jednoduché funkce, jejichž primitivní funkci program nevypočítá.

Bezproblémové integrály

Integrování se provádí pomocí příkazu Integrate. Je nutné vyznačit proměnnou, podle které se integruje. Ve výsledku program nepíše přidané konstanty.

In[23]:=
```
Integrate[Sin[x], x]
```
Out[23]=
$$-\cos(x)$$

In[24]:=
```
Integrate[1 / (x^2 - 1), x]
```
Out[24]=
$$\frac{1}{2}\log(1-x)-\frac{1}{2}\log(x+1)$$

In[25]:=
```
Integrate[1 / (x^2 + 1), x]
```
Out[25]=
$$\tan^{-1}(x)$$

In[26]:=
```
Integrate[Sqrt[1 - x^(2)], x]
```
Out[26]=
$$\frac{1}{2}\left(\sqrt{1-x^2}\;x+\sin^{-1}(x)\right)$$

In[27]:=
```
Integrate[Sqrt[1 + x^(2)], x]
```
Out[27]=
$$\frac{1}{2}\left(\sqrt{x^2+1}\;x+\sinh^{-1}(x)\right)$$

In[28]:=
```
Integrate[x Log[x], x]
```
Out[28]=
$$\frac{1}{2}x^2\log(x)-\frac{x^2}{4}$$

Místo příkazu Integrate lze použít -1.derivaci D[-1]. Pokud dáme místo -1 např. -3, dostaneme třikrát opakovanou integraci.

In[88]:=
```
Derivative[-1][Function[x, x^2]]
```

Out[88]= $x \mapsto \frac{x^3}{3}$

In[89]:=
```
Derivative[-3][Function[x, x^2]]
```

Out[89]= $x \mapsto \frac{x^5}{60}$

Výsledky mírně problémové

Obecně program nevyznačí podmínky, za kterých výsledek platí. V prvním příkladě musí být n≠ -1, ve druhém příkladě je uveden výsledek jen pro x>0 (ale už víme, že *Mathematica* má problémy s absolutní hodnotou). Program tvrdošíjně trvá na výsledku, i když dáme předpoklad, že x je záporné (důvodem je fakt, že program počítá s komplexními funkcemi, jak je vidět z posledního příkladu).

In[29]:=
```
Integrate[x^n, x]
```

Out[29]= $\frac{x^{n+1}}{n+1}$

In[30]:=
```
Integrate[x^(-1), x]
```

Out[30]= $\log(x)$

In[31]:=
```
D[Log[Abs[x]], x]
```

Out[31]= $\frac{\text{Abs}'(x)}{|x|}$

In[32]:=
```
Integrate[x^(-1), x, Assumptions → x < 0]
```

Out[32]= $\log(x)$

In[33]:=
```
Assuming[x < 0, Integrate[1 / x, x]] /. x → -2
```
Out[33]= $\log(2) + i\,\pi$

Někdy se výsledky integrace stejné funkce liší v závislosti na tvaru integrované funkce. Výsledky by se měly lišit o konstantu.

In[34]:=
```
Integrate[2 + (x + 3)^3, x]
```
Out[34]= $\frac{1}{4}(x+3)^4 + 2\,x$

In[35]:=
```
Expand[%]
```
Out[35]= $\frac{x^4}{4} + 3\,x^3 + \frac{27\,x^2}{2} + 29\,x + \frac{81}{4}$

In[36]:=
```
Expand[2 + (x + 3)^3]
```
Out[36]= $x^3 + 9\,x^2 + 27\,x + 29$

In[37]:=
```
Integrate[%, x]
```
Out[37]= $\frac{x^4}{4} + 3\,x^3 + \frac{27\,x^2}{2} + 29\,x$

Problémy

Některé primitivní funkce program vypíše ve tvaru nám neznámém, tj. použije funkce, které jsme nedefinovali. Jsou to případy, kdy se jedná o důležité funkce pro aplikace. První je tzv. chybová funkce erf, druhá sinový integrál Si, třetí Fresnelův sinový integrál.

In[38]:=
```
Integrate[Exp[-x^2], x]
```
Out[38]= $\frac{1}{2}\sqrt{\pi}\,\mathrm{erf}(x)$

In[39]:=

```
Plot[%, {x, 0, 6}]
```

Out[39]=



In[40]:=

```
Integrate[Sin[x] / x, x]
```

Out[40]=

Si(x)

In[41]:=

```
Plot[%, {x, 0, 16}]
```

Out[41]=



In[42]:=
```
Integrate[Sin[x^2], x]
```

Out[42]=

$$\sqrt{\frac{\pi}{2}}\, S\!\left(\sqrt{\frac{2}{\pi}}\, x\right)$$

In[43]:=
```
Plot[%, {x, 0, 6}]
```

Out[43]=



Některé integrály ale program nezná a pak opíše zadání.

In[44]:=
```
Integrate[Log[x] / Cos[x], x]
```

Out[44]=

$$\int \log(x)\sec(x)\,dx$$

Najdou se příklady, kdy je výsledek sice správný na některých intervalech, ale ne na jejich sjednocení. To však program nevypíše. Následující příklad je vybrán z dokumentace programu. Funkce je na intervalu (0,6) spojitá a má tam tedy primitivní funkci (která musí být spojitá, protože má derivaci).

In[45]:=
```
Clear[f]; f[x_] := 1 / (Sin[x] + 2)
```

In[46]:=
```
Plot[f[x], {x, 0, 6}]
```

Out[46]=



Vypočtená primitivní funkce ale spojitá není. Je to proto, že program počítá integrál substitucí, kterou nelze použít v bodě x=π a výsledky na různých intervalech nesrovnává. O tom více v další kapitole.

In[47]:=
```
Clear[h]; h[x_] := Evaluate[Integrate[f[x], x]];
h[x]
```

Out[47]=

$$\frac{2 \tan^{-1}\left(\frac{2 \tan\left(\frac{x}{2}\right)+1}{\sqrt{3}}\right)}{\sqrt{3}}$$

In[48]:=
```
Plot[h[x], {x, 0, 6}]
```

Out[48]=


Integrace funkcí definovaných po částech

Pokud je funkce definována po částech, program ji zintegruje na stejných částech a v bodech styku funkce na jednotlivých částech ,,slepí", takže výsledkem je primitivní funkce na celém definičním oboru. Zintegrované funkce slepí, i když původní funkce v bodech styku spojitá není (viz příklad funkce h1).

In[49]:=
```
Clear[h]; h[x_] := Piecewise[{{-1, x < 1}, {x^2 - 2, x ≥ 1}}]
```

In[50]:=
```
Plot[h[x], {x, -2, 5}, PlotRange → {-1.1, 3.1}]
```

Out[50]=


In[51]:=
```
Integrate[h[x], x]
```

Out[51]=
$$\begin{cases} -x & x \le 1 \\ \frac{x^3}{3} - 2x + \frac{2}{3} & \text{True} \end{cases}$$

In[52]:=
```
Plot[%, {x, -2, 5}]
```

Out[52]=


In[53]:=
```
Clear[h1]; h1[x_] := Piecewise[{{-1, x < 1}, {x^2, x ≥ 1}}]
```

In[54]:=
```
Plot[h1[x], {x, -2, 5}, PlotRange → {-1.1, 3.1}]
```

Out[54]=



In[55]:=
```
Integrate[h1[x], x]
```

Out[55]=

$$\begin{cases} -x & x \leq 1 \\ \frac{x^3}{3} - \frac{4}{3} & \text{True} \end{cases}$$

In[56]:=
```
Plot[%, {x, -2, 5}]
```

Out[56]=



Kupodivu si *Mathematica* neví rady s funkcema definovanýma uvnitř programu, např. funkce signum a absolutní hodnota, i když jsou definovány po částech.

In[57]:=
```
? Sign
```

Sign[x] gives -1, 0 or 1 depending on whether x is negative, zero, or positive. »

In[58]:=
```
Integrate[Sign[x], x]
```

Out[58]=
$$\int \operatorname{sgn}(x)\, dx$$

In[59]:=
```
? Abs
```

Abs[z] gives the absolute value of the real or complex number z. »

In[60]:=
```
Integrate[Abs[x], x]
```

Out[60]=
$$\int |x|\, dx$$

Pokud použijeme basolutní hodnotu jako odmocninu ze čtverce, výsledky budou správné. Problém způsoben výpočtem v komplexních číslech - po přidání podmínky, že se pracuje v reálných číslech, dá program správný výsledek i pro tyto funkce. Další možností je zadefinovat tyto funkce po částech.

In[61]:=
```
Integrate[Sqrt[x^2] / x, x]
```

Out[61]=
$$\sqrt{x^2}$$

In[62]:=
```
Integrate[Sqrt[x^2], x]
```

Out[62]=
$$\frac{x\sqrt{x^2}}{2}$$

In[63]:=
```
Integrate[Sign[x], x, Assumptions → Element[x, Reals]]
```

Out[63]=
$$\begin{cases} -x & x \le 0 \\ x & \text{True} \end{cases}$$

In[64]:=
```
Integrate[Abs[x], x, Assumptions → Element[x, Reals]]
```

Out[64]=

$$\begin{cases} -\frac{x^2}{2} & x \leq 0 \\ \frac{x^2}{2} & \text{True} \end{cases}$$

In[65]:=
```
sign[x_] := Piecewise[{{-1, x < 0}, {1, x > 0}}]
```

In[66]:=
```
Integrate[sign[x], x]
```

Out[66]=

$$\begin{cases} -x & x \leq 0 \\ x & \text{True} \end{cases}$$

In[67]:=
```
abs[x_] := Piecewise[{{-x, x < 0}, {x, x >= 0}}]
```

In[68]:=
```
Integrate[abs[x], x]
```

Out[68]=

$$\begin{cases} -\frac{x^2}{2} & x \leq 0 \\ \frac{x^2}{2} & \text{True} \end{cases}$$

Vrátíme se k příkladu integrace funkce 1/(sin(x)+2), kterou příkaz Integrate nespočítá dobře, protože ji neslepí např. v bodě π (po substituci tg(x/2)). Pokud uvedenou funkci zadefinujeme po částech výsledek je stejný, jestliže ji definujeme i v bodě π (následující funkce p), ale už je správný slepený, jestliže bod π v definici vynecháme (následující funkce p1).

In[69]:=
```
Clear[p];
p[x_] :=
 Piecewise[{{1 / (2 + Sin[x]), x <= Pi},
   {1 / (2 + Sin[x]), x > Pi}}]
```

In[70]:=
```
Plot[p[x], {x, 0, 2 Pi}]
```

Out[70]=



In[71]:=
```
Integrate[p[x], x]
```

Out[71]=

$$\frac{2\tan^{-1}\left(\frac{2\tan\left(\frac{x}{2}\right)+1}{\sqrt{3}}\right)}{\sqrt{3}}$$

In[72]:=
```
Plot[%, {x, 0, 2 Pi}]
```

Out[72]=



In[73]:=

```
Clear[p1];
p1[x_] :=
 Piecewise[{{1 / (2 + Sin[x]), x < Pi},
   {1 / (2 + Sin[x]), x > Pi}}]
```

In[74]:=

```
Plot[p1[x], {x, 0, 2 Pi}]
```

Out[74]=



In[75]:=

```
Integrate[p1[x], x]
```

Out[75]=

$$\begin{cases} \dfrac{2\tan^{-1}\left(\dfrac{2\tan\left(\frac{x}{2}\right)+1}{\sqrt{3}}\right)}{\sqrt{3}} & x \leq \pi \\ \dfrac{2\tan^{-1}\left(\dfrac{2\tan\left(\frac{x}{2}\right)+1}{\sqrt{3}}\right)}{\sqrt{3}} + \dfrac{2\pi}{\sqrt{3}} & \text{True} \end{cases}$$

In[76]:=

```
Plot[%, {x, 0, 2 Pi}]
```

Out[76]=



Existence primitivní funkce ke spojité funkci

V hlavním textu je důkaz existence primitivní funkce ke spojité funkci proveden pomocí aproximace funkcemi po částech lineárními (lomenými čarami). Ukážeme na příkladě, jak tyto aproximace vypadají a jak se blíží k hledané primitivní funkci (v animaci je graf primitivní funkce modtý).

In[77]:=

```
Clear[f, h, g, p]; f[x_] := 6*x^2 - 6*x + 1;
h[x_] := Integrate[f[y], {y, 0, x}];
ColumnForm[{"f=" f[x], "∫f=" h[x],
  Plot[{f[x], h[x]}, {x, 0, 1}, Epilog -> Text[f, {1, 0.65}]]}]
```

Out[78]=

$f = (6x^2 - 6x + 1)$

$\int f = (2x^3 - 3x^2 + x)$



In[79]:=
```
g[x_, n_] :=
  (1 / 2^n) * Sum[f[i / 2^n], {i, 0, Floor[x * 2^n] - 1}] +
   f[Floor[x * 2^n] / 2^n] * (x - Floor[x * 2^n] / 2^n);
p[n_] := Plot[{2 * x^3 - 3 * x^2 + x, g[x, n]}, {x, 0, 1}];
```

In[81]:=
```
Manipulate[Plot[{h[x], g[x, n]}, {x, 0, 1},
  PlotRange → {-0.1, 0.5}], {n, 0, 8, 1}]
```